

# Information Note

## Introduction to Vega dial plans

– Vega 50 6x4 FXS, FXO and BRI & Vega 400



Vega dial plans allow the Vega to route, authenticate and apply number translations to incoming calls in order to onward route them.

This document provides an introduction to constructing dial plan entries to provide the functionality required.

The Annex at the end provides a number of example dial plans.

### **Introduction to Dial Plans**

In a VoIP installation, routing, authentication and number translation can be performed by a SIP proxy, an H.323 gatekeeper, or by rules within the Vega (known as Vega dial plan entries). Vega dial plan entries can in fact be useful even when performing the main routing, authentication and number translation within the SIP proxy or H.323 gatekeeper; the Vega dial planner can be used to, for example, translate local telephony number formats into the standard used in the VoIP network – e.g. for a multi national system, the Vega may pre-pend country code digits onto the caller ID and in-country dialled numbers for calls arriving on the telephony interface, and remove the country code digits on calls from the VoIP network as they are forwarded to the telephony interface. When using the Vega dial planner in this way, the SIP proxy or H.323 gatekeeper only has to deal with fully internationalised numbers, and yet the telephony interface correctly uses local dialling style.

The presentation below provides an overview of the Vega Dial Planner, and explains, using examples, how to write dial plans.



## Dial plans ...

- > Built in Router
  - > List of input conditions and output conditions
- > Plans and Profiles
  - > Up to 200 plans
  - > Up to 25 profiles
  - > Max 50 plans per profile
  - > Profiles enabled / disabled
- > General format:
  - > Source = IF:iiii,further conditions
  - > Destination= IF:jjjj,further details

Vega gateways have a built in routing capability so that based on the physical interface the call arrived on, the telephone number being called, and other information received by the Vega, it can decide how and where to onward route that call. As part of this routing capability the Vega dial planner can specify number translations and other data modifications such as addition or removal of telephone number prefixes. Where the Vega gateway receives Caller ID information, it can also allow or disallow calls based on the phone number that made the call.

The router, also known as the Dial Planner, uses rules based routing - a set of rules, known as dial plan entries are used to specify how calls will be routed. Each Rule (Dial Plan entry) consists of an input condition - which filters which calls are handled by this rule, and a corresponding output specification - which defines how to forward the call.

If an incoming call matches an input condition, the output specification associated with that input condition is used to route the call.

Vega gateways support up to 200 dial plan entries. These dial plan entries can be grouped into profiles. The concept of profiles is to make it easier for the user to manage dial plans. The best way to use profiles is to use them to group dial plan entries with a common theme, e.g. 1 might be used for calls to / from London, the second for calls to / from San Francisco and a third for calls to / from Bangalore. If a change in a dial plan is needed at a later date it is then easier to locate the required dial plan entry to be modified, as it will be one of a few dial plans in the profile, rather than requiring the person updating the dial plans to have to search through all entries (if all dial plans were put in a single profile).

The Vega supports up to 25 profiles, and each profile can contain up to 50 dial plan entries.

Although individual dial plan entries cannot be enabled and disabled, each profile can be enabled or disabled. If a profile is enabled, all the dial plans in that profile will be used alongside all other dial plans in enabled profiles as part of the routing process; if the profile is disabled, all the dial plan entries in that profile will be ignored when the dial planner routes a call.

Profiles do not have any priority associated with them, so if a dial plan entry is in an enabled profile it will be used equally alongside all other dial plans in enabled profiles to decide how to route the call.

Dial plan entries have a general format:

Source = interface (IF:) and the interface ID followed by further conditions that must be met for a call to be handled by this dial plan entry  
Destination = interface (IF:) and the interface ID followed by any further details that are to be sent when onward routing the call.

Source expressions limit the range of incoming calls that are appropriate to this rule (dial plan entry). Destination expressions specify what to do with the call given that it matches the source expression.

If no dial plan entry matches the incoming call information, the call will be dropped.



## Dial Plans – Interface IDs



VegaStream VoIP gateways – Release 8

47


The interface IDs are as shown above; Up until the Vega 6x4, all Vega gateways had interface IDs that were 2 digit IDs; the Vega 6x4 introduced 4 digit interface IDs, where the first two digits indicate the type of interface and the second 2 digits indicate the interface number within that interface type.

Vega 400 gateways running firmware less than R8.0 have 2 digit interface IDs: 01, 02, 03, 04, and 99 or 05. From R8.0 the Vega 400 moves to 4 digit interface IDs: 0401, 0402, 0403, 0404 and 9901 (to 9905) or 0501.

9901 to 9905 (only available from R8.0) select the appropriate SIP profile for routing SIP calls. Although in R8.0 all incoming SIP calls will be received on interface 9901, future builds will separate calls to 9901, 9902 etc depending on their source. Dial plans to handle incoming SIP calls should therefore be written to handle calls from any 99xx interface, i.e. for source expressions use IF:99..



## Dial plans (H.323) ...




VegaStream VoIP gateways – Release 8

48

The information used in dial plan entries is specified as TOKENS. The TOKENS that are appropriate depend on whether it is a dial plan source expression or whether it is a dial plan destination expression that is being specified, also on whether the interface is a telephony interface or a VoIP interface. The slide above shows the valid TOKENS for H.323 Vega gateways, and the slide below shows the valid TOKENS for SIP Vega gateways. As you can see they are almost identical, so once you know how to configure dial plans for SIP, you will also know how to configure dial plans for H.323, and vice-versa. (The greyed 'NAME' TOKEN in the H.323 slide is a TOKEN that is not available in SIP.)



## Dial plans (SIP) ...



VegaStream VoIP gateways – Release 8

49

TOKENS must be in CAPITALS and must be followed by a colon (:)

The TOKENS have the following meanings:

IF: - interface  
 TEL: - telephone number  
 TELC: - telephone number of the caller (Caller ID, Calling Party number)  
 TA: - Transport Address (IP address to send the VoIP call to)<sup>1</sup>  
 TAC: - Transport Address of Calling party (IP address of the device sending the call)  
 DISP: - Display field  
 NAME: - Name used for name dialling  
 NAMEC: - Name of calling party

CAPDESC: - Capability Description set to use ... allows codec selection to be defined on a call by call basis.

QOS: - Quality Of Service profile to use for this call. Needed when 802.1q (VLAN) is used, to specify the VLAN id to use in messages associated with this call

All TOKENS are optional, though it is strongly recommended that IF: is defined for each dial plan entry. Typically beyond that only TEL: and TA: are needed (TA: only on outbound VoIP calls).

### For incoming telephone calls,

IF: defines which interface number(s) this dial plan is appropriate for  
 TEL: defines the dialled telephone number / number range that this dial plan entry handles  
 TELC: not normally used, but if it is, this limits which callers this dial plan entry is appropriate to<sup>2</sup>

<sup>1</sup> This can include an IP port number ':xxxx' e.g. 192.168.1.52:5070

For **outbound VoIP calls**,

IF: will be 05 or 0501 for H.323 and 99 or 9901 to 9905 for SIP.  
TEL: the telephone number to be used for the outbound call  
TA: the IP address (+ IP port) to send the call to (far end destination / SIP proxy / H.323 gatekeeper)  
DISP: the information to present on the display of the receiving device if it has a display  
NAME: where name dialling is used, name can be used instead of TEL: dialling  
TELC: by default the Caller ID that was received in the inbound call will be passed through to the outbound call. If a different Caller ID is required, TELC: allows the outbound caller ID to be overridden  
CAPDESC: Codec Capability Description list – defines which codecs to offer when making this call  
QOS: QOS profile to use – specifically to define which VLAN ID to use when sending out the VoIP call

For **inbound VoIP calls**,

IF: will be 05 or 0501 for H.323 and 99 or 9901 to 9905 for SIP.  
TEL: defines the dialled telephone number / number range that this dial plan entry handles  
TELC: this limits which callers this dial plan entry is appropriate to  
DISP: check that the display information contains a specific value  
TAC: this limits which sending IP devices that this dial plan entry will handle  
NAME: defines the dialled NAME that this dial plan entry handles  
NAMEC: this limits which callers' NAME(s) that this dial plan entry is appropriate to  
QOS: QOS profile to use – specifically to define which VLAN ID to accept calls from

For **outbound telephony calls**,

IF: the ID of a telephony interface  
TEL: the telephone number to be dialled for the outbound call  
DISP: ISDN has the ability to specify the information to present on the display of the receiving device if it has a display; this TOKEN specifies the information to present  
TELC: by default the Caller ID that was received in the inbound call will be passed through to the outbound call. If a different Caller ID is required, TELC: allows the outbound caller ID to be overridden

---

<sup>2</sup> TELC: is very useful when developing new dial plans on a live system. Write new dial plans with a TELC: in the source expression which matches your test telephone's number – only calls from your test telephone will match this source expression and so only calls from that phone will use this dial plan entry when routing calls; calls from all other phones will use the existing dial plan entries. When the new dial plan has been proven to provide the required functionality, the TELC: in the source expression can be deleted making the dial plan available for routing all calls.




## Dial plans ...

- > Regular expressions for source:
  - . a single character
  - < > capture the sequence in < > and store as < n >
  - [ abc ] a single character within the parenthesis
  - [ x-y ] a single character in the range x-y
  - [ ^ abc ] a single character except those within the parenthesis
  - \* the preceding character/expression repeated zero or more times
  - + the preceding character/expression repeated one or more times
  - ? the preceding character/expression repeated zero or one times
  - \ literalise the following character (e.g. \`*` = \* and not a repeat of the previous character)

In order to allow groups of telephone numbers to be routed in a single dial plan Vega gateways support a number of regular expressions (wild cards) that may be used in the **source** part of the dial plan.

As an example, let's consider a company phone system. We will write some dial plan entries for the Vega 400 (SIP) that has incoming calls on interface 0401. (We will assume that dial plan entries are already written in the destination Vega which forwards the calls it receives to the end users via a PBX.)



To start with, let's look at a dial plan on the Vega 400 to route the receptionists number, 01344 784 900, to the other Vega

Source = IF:0401,TEL:01344784900      Destination = IF:9901,TEL:01344784900,TA:w.x.y.z

Calls arriving on interface 0401, which have a telephone number 01344784900 will be handled by this dial plan. These calls will be onward routed thru interface 9901, passing on a telephone number 01344784900, and sending it to IP address w.x.y.z

Now with 350 dial plans we can therefore handle 350 numbers explicitly.

In practice we often want to handle many more numbers than this and these numbers are grouped into sets of numbers. We can use regular expressions to add wild cards to the **source** expressions to allow them to specify groups of numbers.

e.g. consider the company wanting to route the 10-number number group 01344 784 900 to 01344 784 909 to the distant Vega

We can use '.' to indicate that the last digit can be any digit (0 to 9)

Source = IF:0401,TEL:0134478490.

The challenge now is that when onward routing the call we wish to pass on the telephone number that was dialled. If we followed the example above and wrote the TEL: with an explicit number, all calls 01344 784 90x would get sent to 01344 784 900 – not what we want!

We need to be able to indicate to the dial planner that it should use the number that matched the regular expression in the source expression in its destination expression. We do this by using 'angle brackets' '<' and '>' (less than and greater than signs). Angle brackets around numbers and regular expressions in the source expression tell the dial planner that it must store the incoming value that matched this into a local store. Each dial plan entry has its own local stores that start at ID 1 (the dial plan writer does not need to know what other stores other dial plans have used ☺ each dial plan entry starts at store ID = 1).

In the destination expression the stored value is used by putting the store number between angle brackets, e.g. for store 1 use <1>

So we can write our dial plan:

Source = IF:0401,TEL:<0134478490.>      Destination = IF:9901,TEL:<1>,TA:w.x.y.z

Calls arriving on interface 0401, which have a telephone number 0134478490 followed by one further digit will be handled by this dial plan. The digits 0134478490 and the extra digit will be stored in store 1. Calls matching the source expression will be onward routed thru interface 9901, passing on the telephone number stored in store 1 (0134478490 plus the 1 extra digit), sending it to IP address w.x.y.z

e.g. now consider the company expanding and needing to route a 100 number group 01344 784 900 to 01344 784 999

We can use '..' (two dots) to indicate that both the last two digits can be any digit (0 to 9)

Source = IF:0401,TEL:<013447849..>      Destination = IF:9901,TEL:<1>,TA:w.x.y.z

Calls arriving on interface 0401, which have a telephone number 013447849 followed by two further digits will be handled by this dial plan. The digits 013447849 and the extra two digits will be stored in store 1. Calls matching the source expression will be onward routed thru interface 9901, passing on the telephone number stored in store 1 (013447849 plus the 2 extra digits), sending it to IP address w.x.y.z

e.g. now consider the company expanding to a 200 number group 01344 784 800 to 01344 784 999

We can use '..' to indicate that both the last two digits can be any digit (0 to 9); to indicate that the previous digit is an 8 or a 9 we use the set notation [89]

Source = IF:0401,TEL:<01344784[89]..>      Destination = IF:9901,TEL:<1>,TA:w.x.y.z



Calls arriving on interface 0401, which have a telephone number 01344784 followed by an 8 or a 9 and then two further digits will be handled by this dial plan. The digits 01344784, the 8 or the 9 and the two extra digits will be stored in store 1. Calls matching the source expression will be onward routed thru interface 9901, passing on the telephone number stored in store 1 (01344784, the 8 or 9 plus 2 extra digits), sending it to IP address w.x.y.z

The digits between the square brackets indicate the set of numbers that are valid for this digit location in the digit sequence.

For instance

[1246] means that a digit 1 or 2 or 4 or 6 is valid at the point in the digit sequence where the [1246] is written.

There are some other short forms that can be used for writing sets:

e.g.

[5678] ≡ [5-8] ≡ [^123490]

the set 5, 6, 7, 8 is the same as the set 5 to 8, which is the same as the set NOT 1, 2, 3, 4, 9, 0.

Up to this point all the dial plan entries have been a fixed length; we can replace a digit position with 'any digit' or 'one of a set of digits'. It is, however, not always possible to define fixed length dial plan entries, especially for example when handling international calls and the variable length numbering supported in the German network.

The benefit of fixed length dial plans is that the Vega knows exactly when dialling is complete; with variable length dialling the Vega has to run a timer (just as is implemented in all other PBXs and telephony switches) to decide whether the caller has finished dialling. The timer is reset every time a new digit is received. When the timer expires it is deemed that the caller has completed their dialling.

The duration of the timer is a compromise between timing out too early when someone is checking a number in an address book or piece of paper as they dial, and having to wait a long time after dialling before the call connection is completed, due to waiting for this timer to expire. If using variable length dialling ensure that the dtmf timeout period is set appropriately; typically a value of 5 seconds is right.

A typical example of using variable length dialling is a catch-all dial plan that will catch any dialled number, even if it is not one that was expected, and route it through to the operator, e.g.

Source = IF:0401,TEL:.\* Destination = IF:9901,TEL:01344784900,TA:w.x.y.z

The '\*' operates on any digit or regular expression, allowing it to occur 0 or more times, so .\* will match with zero or more digits, each digit being any digit, i.e. any phone number, whatever length.

The operation of the repetition regular expressions can be seen from the following examples:

Operator	*	+	?
Summary	Zero or more times	One or more times	Zero or one time
Example	TEL:123*4	TEL:123+4	TEL:123?4
Matches	124 1234 12334 123....34	1234 12334 123....34	124 1234

The last regular expression is '\', not used often, but sometimes, for instance in call centre applications, telephone numbers are overloaded with 2 or more sets of information; this may be delimited by a \* or #.

If the Vega is to look for 123\*4, then, as seen above, a dial plan saying TEL:123\*4 will not work, what is required is TEL:123\\*4

The backslash turns off the special meaning of \*, so that the Vega really does look for the \* dtmf character.

Regular expressions are valid in any fields in source expressions, including TEL:, IF:, TELC:, TAC:. They are **not** valid in destination expressions – the Vega needs to be told exactly how to route the destination call (see below for valid regular expressions in destination dial plan rules).

Note that white space is NOT allowed in dial plan entries. White space is treated as a terminator and so if found will cause following details to be missed. An exception to this is in names, e.g.in the DISP: field spaces can be used, e.g. DISP:Steve Hight, but in this case the whole dial plan expression must be enclosed in double quotes, e.g.

```
Source = IF:0401,TEL:<013447849..>
          Destination = "IF:9901,TEL:<1>,TA:w.x.y.z,DISP:External Call"
```

When using Variable length dialling, it is possible to configure the Vega to detect a tone that indicates that the dialled number is complete and that it is not necessary to wait (the dtmf timeout) for further digits. By default the Vega is configured to use the DTMF # tone. So typing a # after the dialled number will allow the Vega to forward the call immediately.



## Dial plans ...

- > Regular expressions for destination:
  - ~ FXO only: pause (e.g. waiting for second dial tone)
  - <n> Insert the nth captured sequence from the source expression
  - \ literalise the following character

In destination expressions only 3 types of regular expression may be used. These are <n>, where n=store number ... to carry information from the source expression to the destination expression.

The second is Tilda '~' this inserts a pause in an FXO outdial (only), e.g. for sending a call out from interface 0201 of an FXO:

```
Source = IF:99..,TEL:9<.*>
          Destination = IF:0201,TEL:9~<1>
```

This dial plan uses the leading 9 to know that the call is destined for the PSTN via the PBX to which it is attached. In order to delay the digits to be dialled directly to the PSTN (to allow the PBX to connect to the PSTN dial tone), in the destination expression a delay is added between the 9 and the rest of the number. ('~' works the same way as ';' when setting up modem dialling on a PC).

A '~' inserts a pause = 1 dtmf period (approx 250ms). If more delay is required, multiple '~' characters may be concatenated.

The third is the backslash that literalises certain characters (as described in the source regular expressions)

For further dial plan examples, see Annex 1



## Dial Plans

### > Priorities

- e.g.:

```
IF:01..,TEL:<1234>
```

```
IF:01..,TEL:<123.>
```

```
IF:01..,TEL:<.*>
```

Cost 1 - highest priority (first match)

Cost 2

...

Cost 9 - low priority (first match)

Cost 0 - lowest priority (longest match) ← *default*

When writing dial plans, it is perfectly reasonable that the telephone number ranges handled by different dial plans overlap each other, especially when one of the dial plan entries is a catch-all dial plan.

Fortunately, the Vega has a method of prioritising the dial plans. By default, when a dial plan is created, it is created with a dial plan 'cost' of 0. This causes the Vega to use a 'longest matching' algorithm to decide which is the best dial plan to use in a specific scenario. The more 'explicit digits' defined the longer the match length ascribed to that dial plan entry.

With the dial plans

```
IF:01..,TEL:<1234>
```

```
IF:01..,TEL:<123.>
```

```
IF:01..,TEL:<.*>
```

The longest matching algorithm will look at the TEL: specification and give a match length of 4 explicit digits to <1234>, 3 explicit digits to <123.> and 0 explicit digits to <.\*>

When routing a call the Vega will use the dial plan that has the longest match length and whose source expression matches the call information.

So, if an incoming call arrives on IF:0101 with a dialled number 1234, although all dial plan entries match this incoming call information the longest match is IF:01...,TEL:<1234>, and so this is the dial plan rule that will be used.

If an incoming call arrives on IF:0101 with a dialled number 1235, the first dial plan will not match, but the second two will. The dial plan which matches the incoming call information and has the longest match length is IF:01...,TEL:<123.>, so this is the dial plan rule that will be used.

If an incoming call arrives on IF:0101 with a dialled number which is outside the range 123x, then the only dial plan to match is IF:01...,TEL:<.\*>, so this is the dial plan rule that will be used.

So the Vega makes the 'sensible' choice. The more closely the dial plan definition matches the incoming information, the more likely that dial plan will be used to route the call.

If you wish to override the 'longest match' decisions that the Vega makes, you can do this using the 'cost' parameter.

The lower the cost value, the earlier in the list the dial plan will be for comparison with the incoming call information. 1 is earliest in the list, 2 second, 9 ninth and 0 tenth (last).

In general, because the 'longest match' algorithm makes sensible choices for order, use of 'cost' is seldom needed. Typically it is only required where call representation dial plans are being used. In this case all the source expressions are the same, you need to tell the Vega which order to try the dial plans. Do this by configuring each of the plans with a unique cost.

Costs 1 to 9 do not use the longest matching algorithm to prioritise entries at the same cost, so to get the ordering correct, for every set of overlapping dial plans, only have one dial plan with cost 1, one dial plan with cost 2, ..., one dial plan with cost 9, and then as many dial plans with cost 0 as required.



## Dial plans ...

- > Web browser [Management](#) page and CLI:
  - Show plan
    - Lists all plans
  
- > Web browser [Management](#) page and CLI:
  - Show paths
    - Shows the plans in the priority order
  
- > Using the CLI:
  - Try `<source address tokens>`
    - Test out the plans

Once dial plans have been written, they can be displayed using either the 'show plan', or the 'show paths' command (available on the Command Line Interface and on the Management page of the web browser).

Show plan displays a list of the dial plans in profile / plans-in-profile order.

Show paths can be more useful in that it displays the dial plans in priority order per source interface. 'Show paths' shows the dial plans appropriate to all ports. 'Show paths nnnn' shows the dial plans appropriate to inbound port nnnn.


In order to test dial plan entries without having to make physical phone calls the try command is used. As parameters to the command it takes TOKENS in the same format as used in the dial planner. E.g. to see what would happen if a call arrived on a Vega IF:0103 with a dialled number 1234, type:

```
Try IF:0103,TEL:1234
```

The Vega will display an ordered list of the dial plan entries that would match that incoming call. The first dial plan displayed is the highest priority dial plan entry and this entry will be used to route the call (so long as it is not disabled).



## Advanced Dial Plans



VegaStream VoIP gateways – Release 8

124

If a call fails for a specific reason, the Vega dial planner can be configured to re-present the call to a different destination, e.g. different interface or different telephone number.

For instance if there is a bank of Vega gateways through which calls are made to the PSTN the Vega receiving the initial call will send the call to the first Vega in that bank. The Vega receiving the call (the first Vega in the bank) will then try to make the outbound call on its telephony trunk interface 0401. If that trunk is full (cause code 34 is returned) then that Vega can try interfaces 0402, 0403 and lastly 0404. If all are full (cause code 34 returned from each) then the Vega rejects the call back to the original Vega with cause code 34. That original Vega can then use call re-presentation to try the second Vega in the bank, and so on until the call is either successfully placed – in which case the call proceeds as expected, or all destination trunks are full in which case the original incoming call will be rejected with cause code 34 (Network congested – no channel or circuit available).

This same style of re-presentation can be used to re-present calls to analogue phones connected to FXS ports on a Vega 50, if for example the destination phone is found to be busy (cause code 17). This is useful where multiple users can handle incoming calls, for example on a product support desk.

***For a full list of clear-down cause codes, see the Information Note 'IN\_18 – Q.850 clear-down cause codes'. The easiest way to determine which cause code to use to re-present the call is to watch a log display on as the call fails. The clear-down cause shown in the disconnect message is the cause code to use.***



## Advanced Dial Plans

- > Call re-presentation
  - > Write dial plan for each route
  - > Put dial plans in a group

Plans in this Profile							
Del?	Plan ID	Name	Src	Dest	Cost	Group	Chg?
<input type="checkbox"/>	1	Hunt_Vega_1	IF:040.,TEL:<01344.*>	IF:9901,TEL:<1>,TA:Vega_1_IP	1	1	<a href="#">Modify</a>
<input type="checkbox"/>	2	Hunt_Vega_2	IF:040.,TEL:<01344.*>	IF:9901,TEL:<1>,TA:Veag_2_IP	2	1	<a href="#">Modify</a>
<input type="checkbox"/>	3	Hunt_Vega_3	IF:040.,TEL:<01344.*>	IF:9901,TEL:<1>,TA:Vega_3_IP	3	1	<a href="#">Modify</a>
<input type="checkbox"/>	4	Hunt_Vega_4	IF:0401,TEL:<01344.*>	IF:9901,TEL:<1>,TA:Vega_4_IP	4	1	<a href="#">Modify</a>

- > Set up group with specified clear-down cause(s)

Modify Group	
Group ID	1
Name	default
Cause	0
Active Times	0000-2359
Priority	<input type="checkbox"/>
<input type="button" value="Submit"/>	

To set up call re-presentation,

1. Set up a group (on the dial plan page of the web browser). Set up a name that indicates what the group is for – e.g represent\_on\_busy. Then set up the appropriate cause code, e.g. 17. (If re-presentation is required for multiple cause codes, these can be entered in comma separated list format, e.g. 3,34,41)
2. For each destination that the call needs to be re-presented to, write a dial plan to handle the call arriving and being sent to that destination. (Typically the source expression will be the same for each of these dial plan entries)
3. Configure each of the dial plans to be a member of the group
4. Configure the priority value of each dial plan entry in order to define the order of presentation.

For example, if the gateway is to route calls starting with a prefix 01344 o one of 4 destination IP trunking gateways, the dial plan entries will be:

Source =	Destination =	Group = 1	Cost = 1
IF:04..,TEL:<01344.*>	IF:9901,TEL:<1>,TA:Vega_1_IP		
Source =	Destination =	Group = 1	Cost = 2
IF:04..,TEL:<01344.*>	IF:9901,TEL:<1>,TA:Vega_2_IP		
Source =	Destination =	Group = 1	Cost = 3
IF:04..,TEL:<01344.*>	IF:9901,TEL:<1>,TA:Vega_3_IP		
Source =	Destination =	Group = 1	Cost = 4
IF:04..,TEL:<01344.*>	IF:9901,TEL:<1>,TA:Vega_4_IP		

Where group 1 is configured to re-present calls on cause code 34.

For this Vega to multiple trunking gateway re-presentation, all source expressions are the same, the destination dial plans just differ by the IP address of the destination gateway and the Cost defies the order to try the destination gateways (1 first, 2 second ... etc.)



## Advanced Dial Plans

> Call Presentation groups

- > Sequence Mode:
  - Linear Up
  - Round Robin
  - Random
- > Destinations:
  - list of physical interfaces

Eg: to hunt calls to telephone number '500' over all 8 FXS interfaces:  
Src=IF:99.,TEL:500  
Dest=IF:1001,TEL:<1>

**Call Presentation Group Configuration**

Call Presentation Group 1	
Name	default
Enable	<input type="checkbox"/>
Interface	1001
Sequence Mode	round_robin
Destination Timeout	180
Destination Timeout Action	Try Next Destination
Max Destination Attempts	8
Cause	17
Destinations	IF:0101  IF:0102  IF:0103  IF:0104  IF:0105  IF:0106  IF:0107  IF:0108

Submit

VegaStream VoIP gateways – Release 8

126

From Release 7.5 Call Presentation Groups have been introduced to make re-presenting calls to multiple telephony interfaces even more straightforward. Call presentation groups create a virtual interface that is made up from multiple physical (telephony) interfaces.

Call Presentation Group virtual interfaces **must** only be used in destination expressions. (Regular expressions '.' and '[wxy]' are used in source expressions to specify multiple interfaces.)

To activate a call presentation group, on the dial plan page select Call presentation groups. Add a new entry if required, and select modify.

Firstly in the **Destinations** section enter the physical interface numbers that form this virtual interface. (N.B. they must be physical interfaces and not other virtual interface IDs and not IP interface IDs).

Select enable to enable it. Ensure that the interface number is unique.

Sequence mode defines how the Vega should use the physical ports:

- linear up ... always start at the first listed interface and only use others if calls fail with specified cause code or time out
- round robin ... for each call start at the next interface in the list
- random ... randomly choose a physical interface to start at

Max destination attempts allows the number of interfaces tested to be less than the number defined by the Destinations list (note Vega will only try each interface in the destinations list once even if Max Destination Attempts > the number of interfaces in the Destinations list)

Cause = the cause code or comma separated cause code list that the Vega will use to decide whether to re-present the call to another interface or whether it will clear the call. (This is the same as the cause value used in dial planner groups).

Destination timeout allows a max timeout for trying any physical interface

Destination timeout action specifies what to do if the timeout occurs, either try the next interface or hang up (reject the call).



For FXS type interfaces, the typical configuration will be:

- Destinations will be the interfaces in the 'hunt group'
- cause will be 17 (end user busy)
- destination timeout action will be set up to 'try next interface'
- destination timeout will be set to around 20 seconds

This configuration presents calls to successive interfaces if the call is not answered within the 'destination timeout' time, or if the phone is busy.

For Vega 400 and Vega BRI interfaces (trunking interfaces) the typical configuration will be:

- Destinations will be the interfaces in the 'group' to send the call to; for a Vega between PSTN and PBX 0x01 and 0x03 will be in one CPG and 0x02 and 0x04 will be in a second CPG
- cause will be 34,41 (No circuit / channel available and Temporary failure)
- destination timeout action will be set up to 'hang up'
- destination timeout will be set to around 180 seconds

This configuration presents calls to successive interfaces if there is no circuit / channel available on the trunk or the call receives a 'Temporary Failure' message when the call is presented. If the call is not answered within the 'destination timeout' time the call will clear.



## Advanced Dial Plans

- > Call re-presentation
  - > Write dial plan for each route
  - > Put dial plans in a group

Plans in this Profile							
Del?	Plan ID	Name	Src	Dest	Cost	Group	Chg?
<input type="checkbox"/>	1	Hunt_Vega_1	IF:040.,TEL:<01344.*>	IF:9901,TEL:<1>,TA:Vega_1_IP	1	1	<a href="#">Modify</a>
<input type="checkbox"/>	2	Hunt_Vega_2	IF:040.,TEL:<01344.*>	IF:9901,TEL:<1>,TA:Veag_2_IP	2	1	<a href="#">Modify</a>
<input type="checkbox"/>	3	Hunt_Vega_3	IF:040.,TEL:<01344.*>	IF:9901,TEL:<1>,TA:Vega_3_IP	3	1	<a href="#">Modify</a>
<input type="checkbox"/>	4	Hunt_Vega_4	IF:0401,TEL:<01344.*>	IF:9901,TEL:<1>,TA:Vega_4_IP	4	1	<a href="#">Modify</a>

[Delete](#) [Add](#)

- > Set up group with specified clear-down cause(s)

Modify Group	
Group ID	1
Name	<input type="text" value="default"/>
Cause	<input type="text" value="0"/>
Active Times	<input type="text" value="0000-2359"/>
Priority	<input type="checkbox"/>
<input type="button" value="Submit"/>	

The Group section in the dial plan page can be used for other things as well as call re-presentation.

Active Times enable dial plans when its status is true:

Active times are in an hhmm-HHMM format where hh is the start hour, mm is the start minute, HH is the end hour and MM is the end minute. The dial plans assigned to this group will be enabled if the Vega time is between hhmm and HHMM.

Note, there is no day of week handling – if this complexity is required this should be handled by a proxy / gatekeeper.

#### Priority

Priority is only used for dial plans that route calls to FXO fallback ports (on Vega FXS gateways with 2 FXO fallback ports).

Fallback FXO ports may be used for handling calls to the local PSTN as well as for routing emergency (999, 911, 112, 000 etc.) calls. It is important however, that, even if a local call is being made on an FXO interface, if an emergency call comes in it will be handled. This is achieved by associating the Emergency Call handling dial plan(s) with a Group where the Priority flag is set.

If a call is being handled with a dial plan that is associated with a group where the Priority flag is set, if the existing call is not a 'Priority' call, then the Vega will clear down the non-priority call and will then make the priority call, hence Emergency calls can 'bump' non priority calls.

#### Notes:

1. With Vega gateways calls can be routed from telephony to VoIP, from VoIP to telephony and telephony to telephony; calls from VoIP to VoIP are not supported. (If you need to do VoIP to VoIP calls, loop back an ISDN or CAS trunk.)
2. With Vega 50 Analogue gateways, if the call transfer supplementary service is to be handled on calls between telephone ports on the same Vega gateway, rather than routing the calls directly from one telephony interface to another, the dial plans must be written to send the calls via the LAN interface, i.e. the destination must be:  
`IF:9901,TEL:<1>,TA:<own_IP_address>`  
... and a second dial plan used to route the call from the LAN to the destination telephony interface (this will be needed anyway for calls coming from other IP devices). The SIP stack which handles the call transfer supplementary service only stays in circuit if the call is looped via the LAN – it does not stay active if the call is looped telephony interface to telephony interface directly.  
Calls looped back in this way will be looped in the LAN chip itself; these calls will not reach the wires of the LAN, and so will not affect LAN bandwidth utilisation.
3. How many dial plan entries will I need to write? Typically 1 dial plan entry is needed per telephony interface on the Vega (to handle VoIP to Telephony calls) plus 1 dial plan per IP destination on the LAN (to handle telephony to VoIP calls). More dial plans will be needed if different telephony ports need to handle calls in different ways.

# Annex 1. Example configurations.

The following examples are given below:

1. Examples of Prefix addition and prefix removal
2. Use DNS to allow a single dial plan to handle many IP phones with disparate IP addresses
3. Vega 100 between PSTN and PBX, using VoIP to access 2 SIP phones and a trunking gateway
4. Optional digit 9 for outside line

## 1. Examples of Prefix addition, and prefix removal

Convert an 11 digit number 01344 78 4xxx into a 4 digit extension number 4xxx (from SIP interface to IF:0401)

Source = IF:99..,TEL:0134478<4...>                      Destination = IF:0401,TEL:<1>

Calls arriving on interface 99xx, which have a telephone number 01344784 followed by three further digits will be handled by this dial plan. The digit 4 and the following three digits will be stored away in store 1. Calls matching the source expression will be onward routed thru interface 0401, passing on the telephone number stored in store 1 (the digit 4 plus the three extra digits).

Convert a four digit number 4xxx into an 11 digit number by prefixing it with 01344 78 (from SIP interface to IF:0401)

Source = IF:99..,TEL:<4...>                      Destination = IF:0401,TEL:0134478<1>

Calls arriving on interface 99xx, which have a telephone number 4 followed by three further digits will be handled by this dial plan. The digit 4 and the following three digits will be stored away in store 1. Calls matching the source expression will be onward routed thru interface 0401, passing on the telephone number 01344 78 followed by the digits stored in store 1 (the digit 4 plus the three extra digits).

## 2. Use DNS to allow a single dial plan to handle many IP phones with disparate IP addresses

Where there are many different endpoints at different IP addresses, the normal style of writing dial plans calls for one dial plan for each endpoint, limiting us to about 350 endpoints. A more efficient method – in terms of the number of dial plans needed – is to allow DNS to do the IP address lookup.

The DNS (name to IP address) lookup may be carried out on data stored in the Vega (the lan.hosts table) or on data stored in an external DNS server (ensure that the DNS server's IP address is configured on the LAN settings page). When the Vega finds an IP address that consists of a name, it automatically carries out the DNS lookup; it first tries the lan.hosts table, and if it does not find a match there, it then tries the external DNS server (if the IP address is not 0.0.0.0)

Set the DNS entries to be:


Name	IP
Phone_001	w.x.y.z
Phone_002	a.b.c.d
Phone_nnn	g.h.i.j

Now the dial plan can be written as follows:

```
Source = IF:0401,TEL:<01344784><...> Destination = IF:9901,TEL:<1><2>,TA:Phone_<2>
```

Calls arriving on interface 0401, which have a telephone number 01344 784 followed by three further digits will be handled by this dial plan. The digits 01344784 are stored in store 1 and the following three digits in store 2. Calls matching the source expression will be onward routed thru interface 9901, passing on the whole dialled telephone number (store 1 followed by store 2); the IP address to which the call is sent is the IP address whose DNS name is 'Phone\_' followed by the final three digits of the phone number.

3. Vega 400 between PSTN and PBX, using VoIP to access 2 SIP phones and a trunking gateway



For this configuration, we need one dial plan to send calls to SIP phone 8497, one to send calls to SIP phone 8498, one to send calls to destination 01344 xxx xxx and a catch-all dial plan to send any other calls to the local PSTN. A rule is also needed to route calls from the local PSTN to the PBX.

```

Source = IF:0401,TEL:<8497>           Destination = IF:9901,TEL:<1>,TA:IP_phone_<1>
Source = IF:0401,TEL:<8498>           Destination = IF:9901,TEL:<1>,TA:IP_phone_<1>
Source = IF:0401,TEL:<01344.....>     Destination = IF:9901,TEL:<1>,TA:gateway_01344
Source = IF:0401,TEL:<.*>             Destination = IF:0402,TEL:<1>

Source = IF:0402,TEL:<.*>             Destination = IF:0401,TEL:<1>

```

If required further rules could be written to allow PSTN callers direct dial capability to the SIP phones (via the Vega 400).

#### 4. Optional digit 9 for outside line

With new VoIP systems, the need to dial 9 (or any other digit) for an outside line is typically obsoleted; based on the number dialled it is obvious whether the call is internal or external. In businesses people are used to dialling the 9 (or other digit) for the outside line; at home they would just dial the required telephone number.

In new systems it is therefore useful to allow users to dial the outside telephone number directly, or the outside number prefixed by the digit 9 (or other digit). This can be achieved using the '?' regular expression. An example dial plan is as follows:

```
Source = IF:99..,TEL:9?<.*>          Destination = IF:1001,TEL:<1>
```

Calls arriving on the SIP interface may have an optional 9 followed by any number of digits. The call will be sent out of virtual interface 1001 (defined by a Call Presentation Group), and the telephone number that will be dialled will be the telephone number received from SIP excluding the optional leading digit 9.

Be careful ... consider Emergency numbers that start with the same digit as the digit for an outside line. If they are the same, then ensure explicit rules are written to handle emergency numbers. E.g. for the US emergency number 911 write these rules:

```
Source = IF:99..,TEL:<911>          Destination = IF:1001,TEL:<1>
Source = IF:99..,TEL:9<911>        Destination = IF:1001,TEL:<1>
```

Use show paths to ensure that these two 'emergency' rules take priority over the general purpose rule above.

## Annex 2. Advanced regular expressions.

Vega gateways' regular expressions also support groups ... these extend the way the Vega can use a single dial plan entry to handle multiple different incoming calls. Groups allow, for example, non-consecutive number ranges to be handled in a single dial plan rule.

For example to handle calls from area codes 505, 917 and 614, either 3 dial plan entries can be written:

```
Source = IF:0402,TEL:<505><.*>      Destination = IF:9901,TEL:<1><2>
Source = IF:0402,TEL:<917><.*>      Destination = IF:9901,TEL:<1><2>
Source = IF:0402,TEL:<614><.*>      Destination = IF:9901,TEL:<1><2>
```

Alternatively a single dial plan using groups:

```
Source = IF:0402,TEL:<(505) | (917) | (614)><.*>      Destination = IF:9901,TEL:<1><2>
```

## Annex 2. Advanced tokens.

### Low level ISDN numbering plan, type of number, presentation and screening.

Vega dial plans also support some additional Tokens for low-level control of ISDN called party number and calling party number. These values may only be used in destination dial plan entries, and are only appropriate to Vega 400 and Vega BRI calls:

- PLAN: - Specify the 'Numbering Plan Information' for the called number (TEL:)  
PLANC: - Specify the 'Numbering Plan Information' for the calling number (TEL:)  
TYPE: - Specify the 'Type Of Number' Information for the called number (TEL:)  
TYPEC: - Specify the 'Type Of Number' Information for the calling number (TEL:)  
PRESC: - Specify whether the calling number (TEL:) may be Presented to the called party  
SCRNC: - Specify whether the calling party (TEL:) has been Screened

The following values are accepted:

- PLAN: & PLANC:  
- isdn\_telephony, data, telex, national, private, unknown

- TYPE: & TYPEC:  
- national, international, network\_specific, subscriber, abbreviated, unknown

- PRESC:  
- allowed, not\_available, restricted

- SCRNC:  
- failed<sup>3</sup>, not\_screened, passed, network



### Extended SIP From: field handling

In SIP, the From: field is sometimes extended with additional data, e.g.

```
From: Steve Hight <sip:+441344784917;pn=4917@proxy.com;user=phone>;tag=0082-0000982-05cd
```

In this example

- the Display name is: `Steve Hight`
- the callerID part of the URI is: `+441344784917;pn=4917`

When receiving a call with this From: header, the Vega will populate the TOKENs as follows:

- NAMEC: will contain all digits, letters and any '+'s from the callerID part of the URI
- TELC: will contain all leading digits and '+'s (up to a non digit / non '+' character) from the callerID part of the URI
- DISP: contains the Display name

So for the above example:

- NAMEC: = `+441344784917pn4917`
- TELC: = `+441344784917`
- DISP: = `Steve Hight`

---

<sup>3</sup> 'failed' is not a valid ETSI value (even though it is defined in Q.931)



Contact Details  
Email: [support@vegastream.com](mailto:support@vegastream.com)  
Web: [www.vegastream.com](http://www.vegastream.com)  
[www.vegaassist.com](http://www.vegaassist.com)

EMEA Office  
VegaStream Limited  
The Western Centre  
Western Road  
Bracknell  
Berks RG12 1RW  
UK

+44 (0) 1344 784900

USA Office  
VegaStream Inc.  
6200 Stoneridge Mall Road  
3rd Floor  
Pleasanton  
California 94588  
USA

+1 925 399 6428